	Self-test 1 Solutions


	Unification and lists 1

	 

	 
	Do the following unify and (if so) what are the variable instantiations? 
	

	 
1.
| ?- [Head |Tail] = [brian, mandy, ruth].
Head = brian,

Tail = [mandy,ruth] ? ;

no
 

2.
| ?- [Head | Tail] = [[julie, is, a, wizz]].
Head=[julie,is,a,wizz]

Tail=[] ? ;

no
 

3.
| ?- [who, typed | Tail] = [Head | these, notes, '?'].

In most Prologs, this would give an error message because "these, notes, '?'" isn't within a list. 
  

4.
| ?- [sem242, unification] = [sem242, [Topic]].
no
 

5.
| ?- [sem242, [unification, backtracking, search]] = [Head |Tail].
Head = sem242,

Tail = [[unification,backtracking,search]] ? ;

no
 

6.
| ?- [sem242, [unification, backtracking, search]] = [Head, Tail].
Head = sem242,

Tail = [unification,backtracking,search] ? ;

no
 

7.
| ?- [nova, []] = [Car, Wheels, Tail].
no
 

8.
| ?- [nova] = [Head, Tail].
no
 

9.
| ?- [nova] = [Head |Tail].
Head = nova,

Tail = [] ? ;

no

	Unification and lists 2

	 

	 
	Do the following unify and (if so) what are the variable instantiations? 
	

	 
	 
	

	 
	1. | ?- Var = [1, 2, 3]. 


Yes. Var=[1,2,3] 

2. | ?- Var = [Head, Tail]. 


Yes. Var = [Head,Tail]. 

3. | ?- [_|Tail] = Var. 


Yes. Var = [_A|Tail] 

4. | ?- [Head|Tail] = [Head|Tail]. 


Yes. SICStus Prolog v.3 merely replies true; other Prologs may reply with something like: 
Head = _216, Tail = _217 (where _216 and _217 are both internal variables). 

5. | ?- [Hd1, Hd2|LH_Tail] = [Head|RH_Tail]. 


Yes. Head = Hd1, RH_Tail = [Hd2|LH_Tail] 

Notice here that RH_Tail is given as instantiated to [Hd2|LH_Tail]. 

6. | ?- [LH_Hd1, LH_Hd2, LH_Hd3] = [RH_Hd1, RH_Hd2, RH_Hd3|Tail]. 


Yes. Tail = [], RH_Hd1 = LH_Hd1, RH_Hd2 = LH_Hd2, RH_Hd3 = LH_Hd3 

Notice here that although the left-hand list has three members, it still unifies with the right-hand list (which seems to have four members) because the tail of the list after the list constructor ("|") can be instantiated to the empty list ("[]"). 
  
  

Commentary

Unification of lists is usually kept simple (as with: [a, b, c] = [Head|Tail]). However. examples with more than one head are a source of difficult to both novices and more experienced Prolog programmers. If you wish to use such kinds of examples in your programs, you can always use the Prolog interpreter to test the unifications you believe your program will handle. 

You should not be particularly discouraged if you only got half of the previous questions correct. 



Describe what will happen if you attempt the following and why it will happen: 

1. | ?- S = [a|S]. 

2. | ?- [Hd1, Hd2|Tail] = [Head|Tail]. 

3. | ?- [Hd, Tl] = [Hd|Tl]. 

The key point to remember when looking at these examples is that an uninstantiated variable will become instantiated with the =/2 operator. The first example is the key to the remainder. Here, the variable S is unified with a list of which the same variable S is a part. The instantiation of S becomes infinite. Some Prolog interpreters will simply fail to give any response, but usually they will give something like: 

     S = [a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a,a...

What has happened is that the tail of the list has been instantiated to the whole of the list, which of course includes itself. 

A similar thing happens with the example: [Hd1, Hd2|Tail] = [Head|Tail] where Tail in the right-hand list is required to be instantiated with [Hd2|Tail]. 

The example: [Head, Tl] = [Head|Tl] is apparently more complex. However, it falls into the same kind of pattern, with the variable Tl from the left-hand list being unified with the tail of the second list. The tail of the list is that part after the list constructor ("|") and is always itself a list. So if the Prolog you are using gives any kind of response to such a goal, it would give: 

     Tl = [[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[[...

These examples look particularly horrid. They are given here to illustrate one of the reasons that Prolog programs sometimes fail to give the expected results. 

	Introductory list processing

	 

	 
	 
	

	 
	
| ?- dissect_list([a,b,c]).

a

b

c

[]

yes

| ?- dissect_list([]).

[]

yes

1. | ?- dissect_list([Head|b , c]). 


This gives a syntax error in most Prologs because anything to the right of the list constructor ("|") should be within a single list. The goal dissect_list([Head, b , c]). would not have given an error. 

| ?- dissect_list([Head|[b, c]]).

_70

b

c

[]

true ? ;

no

| ?- dissect_list([1, car(metro, green)]).

1

car(metro,green)

[]

yes

| ?- dissect_list([[a,l,p,h,a], [b,e,t,a]]).

[a,l,p,h,a]

[b,e,t,a]

[]

yes

Note that the argument here is a list of two elements, both of which are lists. 

| ?- dissect_list([Tl|Hd]).

_66

[]

Hd = [] ? ;              {This is the variable instantiation}

_180                   {This is the next alternative}

[]

Hd = [_A] ? ;              {This is the variable instantiation}

_196                   {This is the next alternative}

[]

Hd = [_A,_B] ? ;              {This is the variable instantiation}

_212                   {This is the next alternative}

[]

Hd = [_A,_B,_C] ?                {This is the variable instantiation}

yes
	

	 
Self-test 5, Task 3, Solutions
insert_after/4

This requires one change from the code given for replace/4, namely that both elements should be added to the head of the second list when the first element is found in the first list. 

     /* ************************************************ */

     /*                                                  */

     /*   insert_after/4                                 */

     /*      Arg 1:  Elem1 to be found                   */

     /*      Arg 2:  List1 with Elem1 in                 */

     /*      Arg 3:  Elem2 to be added                   */

     /*      Arg 4:  List2 with Elem2 in                 */

     /*   Summary: True if List2 is List1 with Elem1     */

     /*            followed by Elem2.                    */

     /*   Author: P J Hancox                             */

     /*   Date:   19 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating condition

     insert_after(Elem1, [Elem1|Tail], Elem2, [Elem1, Elem2|Tail]).

     % 2 recursive

     insert_after(Elem1, [Head|Tail1], Elem2, [Head|Tail2]) :-

          insert_after(Elem1, Tail1, Elem2, Tail2).

 

Top of Form

Bottom of Form

Self-test 6, Task 3, Solutions
insert_before_pos/4

This is based on replace_at_pos/4, with the major change being in the terminating condition: 

     /* ************************************************ */

     /*                                                  */

     /*   insert_before_pos/4                            */

     /*      Arg 1:  Counter: integer                    */

     /*      Arg 2:  Elem to be added                    */

     /*      Arg 3:  List1                               */

     /*      Arg 3:  List2 is List1 with Elem added      */

     /*   Summary: is true if List2 is List1 with Elem   */

     /*            before the nth position.              */

     /*   Author: P J Hancox                             */

     /*   Date:   21 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating condition

     insert_before_pos(1, Elem, [Head|Tail], [Elem, Head|Tail]).

     % 2

     insert_before_pos(Pos, Elem, [Head|Tail1], [Head|Tail2]) :-

          Pos1 is Pos - 1,

          insert_before_pos(Pos1, Elem, Tail1, Tail2).

Top of Form

Bottom of Form

	Accumulators

	
     /* ************************************************ */

     /*                                                  */

     /*   delete_nth1/3                                  */

     /*      Arg 1:  Counter: integer                    */

     /*      Arg 2:  List                                */

     /*      Arg 3:  List                                */

     /*   Summary: calls delete_nth/4.                   */

     /*   Author: P J Hancox                             */

     /*   Date:   22 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     delete_nth1(Pos, List1, List2) :-

          delete_nth1(1, Pos, List1, List2).

     /* ************************************************ */

     /*                                                  */

     /*   delete_nth1/4                                  */

     /*      Arg 1:  Counter: integer                    */

     /*      Arg 2:  Position: integer                   */

     /*      Arg 3:  List with element                   */

     /*      Arg 4:  List without element                */

     /*   Summary: true if Arg 3 is Arg 4 without Elem   */

     /*            at Pos.                               */

     /*   Author: P J Hancox                             */

     /*   Date:   22 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating condition

     delete_nth1(Pos, Pos, [Elem|Tail], Tail).

     % 2 recursive

     delete_nth1(Cnt, Pos, [Head|Tail1], [Head|Tail2]) :-

          Cnt1 is Cnt + 1,

          delete_nth1(Cnt1, Pos, Tail1, Tail2).

	Correcting min/3

	
     /* ************************************************ */

     /*                                                  */

     /*   min/3                                          */

     /*      Arg 1:  number                              */

     /*      Arg 2:  number                              */

     /*      Arg 3:  number                              */

     /*   Summary: Arg3 is the minimum of Arg 1 and      */

     /*            Arg 2.                                */

     /*   Author: P J Hancox                             */

     /*   Date:   26 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     min(Min, Max, Min) :-

          Min < Max.

     min(Max, Min, Min) :-

          Min < Max.

This is easily corrected by the addition of one test. Note that this solution doesn't have recourse to extra-logical extensions of Prolog, such as the cut. 

	Multiple solutions in list processing

	
     /* ************************************************ */

     /*                                                  */

     /*   delete_all/3                                   */

     /*      Arg 1:                                      */

     /*      Arg 2:  List                                */

     /*      Arg 3:  List                                */

     /*   Summary: true if Arg 3 is Arg 2 with all       */

     /*            instances of Arg 1 deleted.           */

     /*   Author: P J Hancox                             */

     /*   Date:   28 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating condition

     delete_all(_, [], []).

     % 2 recursive - Elem matches head

     delete_all(Elem, [Elem|Tail1], Tail2) :-

          delete_all(Elem, Tail1, Tail2).

     % 3 recursive - Elem doesn't match head

     delete_all(Elem, [Head|Tail1], [Head|Tail2]) :-

          \+ (Elem = Head),

          delete_all(Elem, Tail1, Tail2).

The terminating condition is now the empty list because we need to work through the entire list to find all elements. There are two recursive conditions: one for when Elem unifies with the head of the list and a second for when Elem and the Head of the list don't unify. 

	Unification and equality again

	
     /* ************************************************ */

     /*                                                  */

     /*   delete_all_eq/3                                */

     /*      Arg 1:                                      */

     /*      Arg 2:  List                                */

     /*      Arg 3:  List                                */

     /*   Summary: true if Arg 3 is Arg 2 with all       */

     /*            instances of Arg 1 deleted.           */

     /*   Author: P J Hancox                             */

     /*   Date:   28 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating condition

     delete_all_eq(_, [], []).

     % 2 recursive - Elem matches head

     delete_all_eq(Elem, [Head|Tail1], Tail2) :-

          Elem == Head,

          delete_all_eq(Elem, Tail1, Tail2).

     % 3 recursive - Elem doesn't match head

     delete_all_eq(Elem, [Head|Tail1], [Head|Tail2]) :-

          Elem \== Head,

          delete_all_eq(Elem, Tail1, Tail2).

	Top of Form

	Templates, determinism and non-determinism

	
     /* ************************************************ */

     /*                                                  */

     /*   memb_ext/4                                     */

     /*      Arg 1: Element                              */

     /*      Arg 2: List                                 */

     /*      Arg 3: Atom, being a relation.              */

     /*      Arg 4: Atom, either det or non.             */

     /*   Summary: true if Arg 3 is an atom and          */

     /*            memb_ext1/4 is true.                  */

     /*   Author: P J Hancox                             */

     /*   Date:   28 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1

     memb_ext(Elem, List, Relation, Det) :-

          atom(Relation),

          memb_ext1(Elem, List, Relation, Det).

     /* ************************************************ */

     /*                                                  */

     /*   memb_ext1/4                                    */

     /*      Arg 1: Element                              */

     /*      Arg 2: List                                 */

     /*      Arg 3: Atom, being a relation.              */

     /*      Arg 4: Atom, either det or non.             */

     /*   Summary: true if Arg 1 and the head of Arg2    */

     /*            are related by Arg 3. Works deter-    */

     /*            minstically if Arg 4 is det; non-     */

     /*            determinstically if Arg4 is non.      */

     /*   Author: P J Hancox                             */

     /*   Date:   28 October 1994                        */

     /*                                                  */

     /* ************************************************ */

     % 1 terminating

     memb_ext1(Elem, [Head|_], Relation, _) :-

          functor(Relation1, Relation, 2),

          arg(1, Relation1, Elem),

          arg(2, Relation1, Head),

          call(Relation1).

     % 2 recursive - deterministic

     memb_ext1(Elem, [Head|List], Relation, det) :-

          functor(Relation1, Relation, 2),

          arg(1, Relation1, Elem),

          arg(2, Relation1, Head),

          \+ call(Relation1),

          memb_ext1(Elem, List, Relation, det).

     % 3 recursive - non-deterministic

     memb_ext1(Elem, [_|List], Relation, non) :-

          memb_ext1(Elem, List, Relation, non).

The main difference is the splitting the recursive clause into two. If the parameter is 'det' then the failure of the relation has to be tested (ie in the sub-goal \+ call(Relation1)); if the parameter is 'non' then there is no need to test the relationship. 

